Событие нажатие, Асинх.
Создал простую конфигурацию. Один справочник, в нем один реквизит "Данные картинки".
Сделал процедуры

&НаКлиенте
Асинх Процедура КартинкаНажатие(Элемент, СтандартнаяОбработка)
 СтандартнаяОбработка = Ложь;
 ЗагрузитьКартинку(Неопределено);
КонецПроцедуры

#Область ПрограммныйИнтерфейс

&НаКлиенте
Асинх Процедура ЗагрузитьКартинку(Команда)
 ПараметрыДиалога = Новый ПараметрыДиалогаПомещенияФайлов("Выберите изображение", Ложь, "Изображения | *.jpg; *.png");
 Обещание = ПоместитьФайлНаСерверАсинх(,,,ПараметрыДиалога, УникальныйИдентификатор);
 ВыбранныйФайл = Ждать Обещание;
 Если ВыбранныйФайл <> Неопределено Тогда
  СсылкаНаКартинку = ВыбранныйФайл.Адрес;
  Модифицированность = Истина;
 КонецЕсли;
КонецПроцедуры

#КонецОбласти

При нажатии именно на поле картинки появляется окошко с адресом хранилища. Как сделать чтобы оно не появлялось?
(Через кнопку "Загрузить картинку" окошко не появляется.)
---
Т.е. в Асинх обработчиках событий установка СтандартнаяОбработка и Отказ не работает. Вот ответ от 1С почему так, правда там про другой обработчик события, но суть та же:

Это известный эффект, который при первой встрече с ним часто оказывается неожиданным.
Вспомним, что все параметры в Асинх процедуре/функции всегда передаются по значению.
В то же время, обработчик формы ПередЗакрытием() путем изменения значений параметров информирует вызывающую сторону о том,
что нужно сделать после выполнения обработчика. Например, отказаться от закрытия формы.
В результате, если сделать обработчик ПередЗакрытием() Асинх, то изменения значений параметров <Отказ>, <ТекстПредупреждения> или <СтандартнаяОбработка> не будут видны вызывающей стороне.
Таким образом, обработчики событий, которые должны что-то сообщить вызывающей стороне, нельзя делать Асинх.
---
Убрать Асинх у процедуры "КартинкаНажатие", в данном случае достаточно что ЗагрузитьКартинку() асинхронная
---
Переписал процедуру КартинкаНажатие вот так:
&НаКлиенте
Асинх Процедура КартинкаНажатие(Элемент, СтандартнаяОбработка)
 СтандартнаяОбработка = Ложь;
 ПараметрыДиалога = Новый ПараметрыДиалогаПомещенияФайлов("Выберите изображение", Ложь, "Изображения | *.jpg; *.png");
 Обещание = ПоместитьФайлНаСерверАсинх(,,,ПараметрыДиалога, УникальныйИдентификатор);
 ВыбранныйФайл = Ждать Обещание;
 Если ВыбранныйФайл <> Неопределено Тогда
  СсылкаНаКартинку = ВыбранныйФайл.Адрес;
  Модифицированность = Истина;
 КонецЕсли;
КонецПроцедуры

Чтобы точно внутри был вызов асинхронности. Окошко всё равно есть.
---
Т.е. в Асинх обработчиках событий установка СтандартнаяОбработка и Отказ не работает. Вот ответ от 1С почему так, правда там про другой обработчик события, но суть та же:

Это известный эффект, который при первой встрече с ним часто оказывается неожиданным.
Вспомним, что все параметры в Асинх процедуре/функции всегда передаются по значению.
В то же время, обработчик формы ПередЗакрытием() путем изменения значений параметров информирует вызывающую сторону о том,
что нужно сделать после выполнения обработчика. Например, отказаться от закрытия формы.
В результате, если сделать обработчик ПередЗакрытием() Асинх, то изменения значений параметров <Отказ>, <ТекстПредупреждения> или <СтандартнаяОбработка> не будут видны вызывающей стороне.
Таким образом, обработчики событий, которые должны что-то сообщить вызывающей стороне, нельзя делать Асинх.
---
Вот спасибо.